
 

CONTRAST、およびESTIMATEステートメントを記述す

る例題 1 

 
改変履歴 ...............................................................................................................................2 
はじめに ...............................................................................................................................3 
例題 1：交互作用項を含む 2 要因の分散分析.......................................................................4 

ESTIMATEステートメントを用いてセル平均を計算する ...............................................5 
平均値の差を推定、および検定する ...............................................................................10 
より複雑な対比の指定 .................................................................................................... 11 
ある一つの交互作用項の平均と、交互作用項全ての平均との比較 ................................13 

例題 2：交互作用項を含む 3 要因の分散分析.....................................................................14 
例題 3：交互作用項を含む 2 要因のロジスティック回帰モデル ........................................17 

Dummyコーディング .....................................................................................................18 
Dummyコーディングを用いてオッズ比の推定と検定を行う .....................................19 
枝分かれ（ネステッド）モデル(Nested Model) .........................................................23 

Effectsコーディング .......................................................................................................25 
Effectsコーディングによるダミー変数を用いて、オッズ比の推定と検定を行う.......26 
Effectコーディングに基づいて、より複雑なCONTRASTステートメントの指定を行う

....................................................................................................................................28 
例題 4. モデルの比較 – 尤度比検定.............................................................................31 

尤度比検定を構成する ....................................................................................................34 
モデルを比較するために「対比」を記述する.................................................................36 

付録 ロジスティック回帰の参考文献.................................................................................38 

                                                  
1 この FAQ は、米国 SAS Institute Inc.の Web ページに記載されている内容をもとに作成

しています。 
http://support.sas.com/kb/24/447.html 

 
使用している SAS プログラムは、下記の米国 SAS Institute Inc.の Web ページから入手す

ることができます。 
http://support.sas.com/kb/24/addl/fusion24447_1_contrasts.sas.txt 
 
また、SAS 出力は全てオリジナルと同様に英語版 SAS の出力です。日本語版 SAS でプロ

グラムを実行したときは、当該出力を適宜読みかえてください。 

http://support.sas.com/kb/24/447.html
http://support.sas.com/kb/24/addl/fusion24447_1_contrasts.sas.txt


改変履歴 

初版        2007/05/17 
第 2 版      2007/12/25 



 

はじめに 

 
“group 1における処置Aの効果と group 2における処置Aの効果は同等であるか”といった

仮説を適切な方法で検定する場合、当てはめモデル(fitted model)を使用した数学的な仮説

へ正しく置き換えることが極めて重要です。この置き換えを手助けすることは、SAS テク

ニカルサポートから提供されるサービスの範囲外となりますが、以下の議論や例が参考と

なりますと幸甚です。なお、SASテクニカルサポートでは、CONTRAST、およびESTIMATE
ステートメントに関する構文などの問合わせについてサポートを提供しています。 
 
以下の例からも、CONTRAST や ESTIMATE ステートメントを適切に記述するためには、

いくつかの重要なステップがあります。 
 
• プロシジャを使って当てはめモデルを記述する。ここでは以下の 2 点が特に肝要です。 
 

パラメータ化 
CLASS ステートメントによって生成されるデザイン変数が、どのようにコード化されてい

るか。“indicator(または dummy)”コーディングと、“effects(または、deviation from mean、
「平均からの偏差」)”コーディングの 2 つが、SAS において最もよく使用されるパラメー

タ化の方法です。 
 

パラメータの順序 
(交互作用項など)複数のパラメータが含まれる効果でのパラメータの順序。この順序は、

CLASS ステートメントで指定されている変数の順序や、PROC または CLASS ステートメ

ントの ORDER=オプションの設定に依存します。プロシジャが表示するパラメータ推定値

のテーブルから、パラメータの順序を確認することができます。 
 
 
• モデルのパラメータを用いて、検定における仮説や推定する統計量を書き下し、簡略

にします。対比の場合には、「contrast = 0」 の形式で帰無仮説を記述して、次に数式

の左辺を簡略化します。たとえば 2つの平均を比較する場合、帰無仮説をμ1 - μ2 = 0 と
与え、μ1 - μ2 の部分をモデルのパラメータで記述することになります。 

 
• パラメータに対する乗数を係数として、CONTRAST や ESTIMATE ステートメントを

記述します。このとき、プロシジャにおけるモデルのパラメータと同じ順序になるよ



うに注意する必要があります。 

例題 1：交互作用項を含む 2 要因の分散分析 

 
5 水準の因子 A と 2 水準の因子 B からなるモデルを考えます。 
 
 Yijk = μ + αi + βj + αβij +εijk,   i=1,2, …,5,  j=1,2,  k=1,2,…,nij 
 
ここで、εijkは独立かつ平均が 0である正規分布に従うと仮定します。以下のプログラムは、

このモデルに従ったサンプルデータを作成するものです。 

 
data test; 

   seed=6342454; 

   do a=1 to 5; 

      do b=1 to 2; 

         do rep=1 to ceil(ranuni(seed)*5)+5; 

            y=5 + a + b + a*b + rannor(seed); 

            output; 

         end; 

      end; 

   end; 

run; 

 
 
次のプログラムでは、主効果と交互作用項を含んだモデルの推定を行っています。

LSMEANSステートメントを指定することによって、この例におけるA*Bという項の 10
個のセルに対するセル平均（LS平均、最小 2 乗平均）を算出しています。Eオプションを

使うと、LS平均の計算で使用された係数ベクトルが表示され、また各セル平均がどのよう

に構成されたかを確認できます。 
 
proc mixed data=test; 

   class a b; 

   model y=a b a*b / solution; 

   lsmeans a*b / e; 

run; 

 



Least Squares Means 

Effect a b Estimate Standard Error DF t Value Pr > |t| 

a*b 1 1 7.5193 0.2905 74 25.88 <.0001 

a*b 1 2 10.0341 0.2598 74 38.62 <.0001 

a*b 2 1 10.4189 0.2739 74 38.04 <.0001 

a*b 2 2 12.5812 0.3355 74 37.50 <.0001 

a*b 3 1 11.5853 0.3355 74 34.54 <.0001 

a*b 3 2 15.7347 0.2598 74 60.55 <.0001 

a*b 4 1 14.5552 0.3355 74 43.39 <.0001 

a*b 4 2 19.3499 0.2598 74 74.47 <.0001 

a*b 5 1 16.3459 0.2739 74 59.68 <.0001 

a*b 5 2 21.6220 0.2598 74 83.21 <.0001 

 
 
ESTIMATEステートメントを用いてセル平均を計算する 
 
各セル平均は、ESTIMATE ステートメントを使用して、対応するモデルパラメータの線形

結合として得ることもできます。AB11 セルと AB12 セルに対する平均(上のテーブルで黄色

で表示されている部分)は、ESTIMATE ステートメントを用いて以下のように求めること

ができます。ESTIMATE ステートメントで必要となる係数は、推定したい統計量を当ては

めモデルに基づいて記述することによって決定されます。前述のモデル(1)の場合では、

AB12 セルの平均 μ12は以下のように記述できます。 
μ12 = (ΣkY12k)/n12  

= (Σkμ)/n12 + (Σkα1)/n12 + (Σkβ2)/n12 + (Σkαβ12)/n12 + (Σkε12k)/n12  
 

誤差項(εijk)の平均は 0 として仮定するので、この式は次のようになります。 
 

=μ + α1 + β2 + αβ12  



同様に、AB11 セルの平均に対しては以下のように記述されます。 
μ11 = μ + α1 + β1 + αβ11  

 
このことから、AB12 セルの平均を推定するためには μ, α1, β2, と αβ12の推定値を足しあわ

せる必要があります。これは、パラメータ推定値のベクトル(solution ベクトル)と係数ベク

トルを掛けることによって計算されます。MIXED プロシジャにおける solution ベクトル

は、MODELステートメントで SOLUTIONオプションを指定して実行することによって、

“固定効果の解(Solution For Fixed Effects)”テーブルの“推定値(Estimate)”の列に表示

されます。 
 

Solution for Fixed Effects 

Effect a b Estimate Standard Error DF t Value Pr > |t| 

Intercept     21.6220 0.2598 74 83.21 <.0001 

a 1   -11.5879 0.3675 74 -31.53 <.0001 

a 2   -9.0408 0.4243 74 -21.31 <.0001 

a 3   -5.8874 0.3675 74 -16.02 <.0001 

a 4   -2.2722 0.3675 74 -6.18 <.0001 

a 5   0 . . . . 

b   1 -5.2762 0.3775 74 -13.97 <.0001 

b   2 0 . . . . 

a*b 1 1 2.7613 0.5426 74 5.09 <.0001 

a*b 1 2 0 . . . . 

a*b 2 1 3.1139 0.5745 74 5.42 <.0001 

a*b 2 2 0 . . . . 

a*b 3 1 1.1268 0.5680 74 1.98 0.0510 



a*b 3 2 0 . . . . 

a*b 4 1 0.4815 0.5680 74 0.85 0.3994 

a*b 4 2 0 . . . . 

a*b 5 1 0 . . . . 

a*b 5 2 0 . . . . 

 
 
このモデルに対しては、パラメータ推定値に相当する solution ベクトルは 18 の要素からな

ります。最初の要素は切片 μに対するパラメータ推定値です。次の 5 つの要素は因子 A の

各水準 α1から α5に対するパラメータ推定値、その次の 2 つの要素は効果 B の各水準 β1と

β2、に対するパラメータ推定値、最後の 10 個の要素は A*B の交互作用項 αβ11 から αβ52に

対するパラメータ推定値となります。 
 
次に、solutionベクトルと掛け合わせる係数ベクトル(同じく 18個の要素)を選び出します。

切片(μ)に対しては係数として 1、α1に対する推定値を取り上げるために効果 A に対しては

係数として(1 0 0 0 0)、β2 に対する推定値を取り上げるために効果 B に対しては係数とし

て(0 1)、αβ12 に対する推定値を取り上げるために交互作用項 A*B に対しては係数として(0 
1 0 0 0 0 0 0 0 0)をそれぞれ選びます。 
 
ESTIMATE ステートメントでは、上記の係数ベクトルを各効果に対して指定します。 
 
  estimate 'AB12' intercept 1 

                                 a 1 0 0 0 0 

                                 b 0 1 

                               a*b 0 1 0 0 0 0 0 0 0 0; 

 
 
上記と同じ係数ベクトルが、LSMEANSステートメントでEオプションを指定することによ

って得られる“a*b の最小 2 乗平均に対する係数” (Coefficients for a*b Least Squares 
Means)の表と同じであることに留意してください。この表における 0 は空欄となっていま

す。Row2 がAB12 セルの平均を計算するための係数ベクトルであることに注目しましょう。 
 
 



 

Coefficients for a*b Least Squares Means 

Effect a b Row1 Row2 Row3 Row4 Row5 Row6 Row7 Row8 Row9 Row10

Intercept     1 1 1 1 1 1 1 1 1 1

a 1   1 1           

a 2       1 1         

a 3         1 1       

a 4           1 1    

a 5              1 1

b   1 1   1  1  1   1  

b   2   1  1  1  1   1

a*b 1 1 1             

a*b 1 2   1           

a*b 2 1     1          

a*b 2 2      1         

a*b 3 1       1        

a*b 3 2        1       

a*b 4 1         1      

a*b 4 2          1    

a*b 5 1            1  

a*b 5 2              1

 



効果のパラメータ推定値において、変数の水準がどのような順番であるかが重要となりま

す。例えば、A*B 交互作用項に対するパラメータ推定値において、2 番目の推定値は αβ12,
の推定値に相当します。これは、変数 A の水準が変動する前に、変数 B の水準が変動する

からです。仮に、CLASS ステートメントにおいて変数 A の前に変数 B が指定されている

場合には、2 番目の推定値は αβ21 に対する推定値となります。この場合、αβ12 の推定値は

A*B 交互作用項における 6 番目の推定値に相当しますので、ESTIMATE ステートメントに

おいて係数ベクトルの指定を変更する必要があります。 
 
AB11 セルの平均に対するESTIMATEステートメントは、パラメータ推定値の対応している

線形式となるように、セル平均をモデルパラメータにて記述することによって、上記と同

じように推定することになります。その係数は、“a*b の最小  2 乗平均に対する係

数”(Coefficients for a*b Least Squares Means)の表におけるRow1 と同じとなります。次の

サンプルコードでは、モデルを推定し、LSMEANSステートメントを用いて各セル平均を

算出しています。また、対応する係数をESTIMATEステートメントにて指定し、AB11、AB12

セルの平均を算出しています。 
 
proc mixed data=test; 

   class a b; 

   model y=a b a*b; 

   lsmeans a*b; 

   estimate 'AB11' intercept 1 a 1 0 0 0 0 b 1 0 

                               a*b 1 0 0 0 0 0 0 0 0 0; 

   estimate 'AB12' intercept 1 a 1 0 0 0 0 b 0 1 

                               a*b 0 1 0 0 0 0 0 0 0 0; 

run; 

 
 

Estimates 

Label Estimate Standard Error DF t Value Pr > |t| 

AB11 7.5193 0.2905 74 25.88 <.0001 

AB12 10.0341 0.2598 74 38.62 <.0001 

 
 



平均値の差を推定、および検定する 
 
ここでは、AB11 と AB12 のセル平均が等しいかどうかを検定するものとします。このとき、

検定する帰無仮説は以下のようになります。 
H0: μ11 = μ12 

 
または、このようにも記述できます。 

H0: μ11 – μ12 = 0 
 
モデルパラメータを使用すると、この対比を次の通りに記述します。 
 

μ11 – μ12 = (μ + α1 + β1 + αβ11) – (μ + α1 + β2 + αβ12) = β1 – β2 + αβ11 – αβ12 
 
この式では切片と効果 A に対する係数は打ち消され、最後の係数ベクトルからこれらの効

果は除外されていることに留意してください。ただし、交互作用項 A*B に対する係数に加

え、効果 B に対する係数は残っています。平均の差が 0 であるかを検定するために、

CONTRAST ステートメントでこれらの係数を使用しましょう。このとき、プロシジャにお

けるモデルパラメータの順序と係数の順序が同じとなるように注意する必要があります。 
 
CONTRAST ステートメントの結果は、ESTIMATE ステートメントを使用して再現するこ

ともできます。ESTIMATE ステートメントでは、セル平均の差の推定値(-2.5148)とこの差

が 0 であるかに対する t 検定の結果が表示されますが、その一方で CONTRAST ステート

メントでは差が 0 であるかに対する F 検定の結果が出力されることに注意してください。

これらの検定は同等なものとなります。このようなシンプルなペアごとの比較の場合、

LSMEANS ステートメントで PDIFF オプションを指定することによっても同じ結果を得

ることができます。 
 
次のプログラムでは、対比に対するこれら 3 通り全ての方法を行っています。 
 
proc mixed data=test; 

   class a b; 

   model y= a b a*b; 

   lsmeans a*b / pdiff; 

   contrast 'AB11 - AB12' b 1 -1 

                               a*b 1 -1 0 0 0 0 0 0 0 0; 

   estimate 'AB11 - AB12' b 1 -1 



                               a*b 1 -1 0 0 0 0 0 0 0 0; 

run; 
 
 
以下は、CONTRAST、ESTIMATE、および LSMEANS ステートメントによる結果で、互

いに関連している部分の出力です。 
 

Estimates 

Label Estimate Standard Error DF t Value Pr > |t| 

AB11 - AB12 -2.5148 0.3898 74 -6.45 <.0001 

 

Contrasts 

Label Num DF Den DF F Value Pr > F

AB11 - AB12 1 74 41.63 <.0001

 

Differences of Least Squares Means 

Effect a b _a _b Estimate Standard 

Error

DF t Value Pr > |t| 

a*b 1 1 1 2 -2.5148 0.3898 74 -6.45 <.0001 

 
 
より複雑な対比の指定 
 
ここでは、AB11 と AB12 セルの平均と、AB21 と AB22 セルの平均の検定を行うものとしま

す。このとき、帰無仮説は以下のようになります。 
 

H0: 1/2 (μ11 + μ12) = 1/2 (μ21 + μ22) 
 
AB11 と AB12 セルの平均の推定値に対する係数は、モデル式から決められます。それぞれ



AB11、AB12 セルの平均は、 
 

μ11 = μ + α1 + β1 + αβ11 
μ12 = μ + α1 + β2 + αβ12 

 
となり、この 2 つのセル平均の平均は、以下のように記述することができます。 
 

½(μ11+μ12)  =  ½[ (μ + α1 + β1 + αβ11) + (μ + α1 + β2 + αβ12) ]   
                      = μ + α1 + ½β1 + ½β2 + ½αβ11 + ½αβ12  

 
AB21 と AB22 セルの平均に対する係数も同様に求めることができます。そこで、前と同じ

ように、2 つの係数ベクトルの差を求めることによって、2 つの平均差を検定するための係

数ベクトルを得ることができます。これらの係数を、以下の ESTIMATE ステートメント、

および CONTRAST ステートメントにて指定することになります。なお、一つの効果に対

する係数では、後ろに続く 0 は指定しなくても構いません。 
 
proc mixed data=test; 

   class a b; 

   model y= a b a*b; 

   estimate 'avg AB11,AB12' intercept 1 a 1 b .5 .5 

                                 a*b .5 .5; 

   estimate 'avg AB21,AB22' intercept 1 a 0 1 b .5 .5 

                                 a*b 0 0 .5 .5; 

   contrast 'avg AB11,AB12 - avg AB21+AB22' a 1 -1 

                                                    a*b .5 .5 -.5 -.5; 

run; 

 

Estimates 

Label Estimate Standard Error DF t Value Pr > |t| 

avg AB11,AB12 8.7767 0.1949 74 45.04 <.0001 

avg AB21,AB22 11.5001 0.2165 74 53.11 <.0001 

 



Contrasts 

Label Num DF Den DF F Value Pr > F 

avg AB11,AB12 - avg AB21+AB22 1 74 87.39 <.0001 

 
 
ある一つの交互作用項の平均と、交互作用項全ての平均との比較 
 
ここでは、変数 A は 2 水準、変数 B は 3 水準とし、AB12 セルの平均が全 6 個のセル平均

の平均と違いがあるかを検定することにします。このとき、帰無仮説は以下のように記述

されます。 
 

H0: μ12 – 1/6 Σijμij = 0 
 
モデルでは、前記の式(1)における j の取り得る値のみが変更されています。モデルのパラ

メータを用いて、AB12 セルの平均は以下のように記述することができます。 
 

μ12 = μ + α1 + β2 + αβ12 
 
同じくモデルのパラメータを用いて、6 つのセル平均の平均は、以下のように記述されます。 
 

Σijμij = 1/6 ΣiΣj(μ + αi + βj + αβij) = μ + 3/6 Σiαi + 2/6 Σjβj + 1/6 ΣiΣjαβij 
 
つまり、変数 A のパラメータに対する係数は 1/2、B に対しては 1/3、A*B に対しては 1/6
となります。ただし、これらの係数から、 
 
estimate 'avg ABij' intercept 1 a .5 .5 b .333 .333 .333 

                                      a*b .167 .167 .167 .167 .167 .167; 

 
と記述した場合、プロシジャはこの ESTIMATE ステートメントに対する結果を算出せず、

Non-Est と表示されるか、ログ画面に以下のメッセージが表示されます。 
 
NOTE: avg ABij is not estimable. 

 
係数である 1/3、1/6 が厳密に指定されていないことから、推定可能ではない(nonestimable)



と表示されています。この問題を回避するためには、DIVISOR=オプションを用います。

このオプションを使用すると、ESTIMATE ステートメントにおける全ての係数の分母に相

当する値を指定することができます。 
 
最後に、これらの対比における係数を用いて、帰無仮説における μ12 – Σijμij i の部分をモデ

ルのパラメータで記述します。切片に対しては 0、変数 A に対しては(1/2、-1/2)、変数 B
に対しては(–1/3, 2/3, –1/3)、 交互作用 A*B に対しては(-1/6, 5/6, -1/6, -1/6, -1/6, -1/6 )とい

う係数となります。次のプログラムはモデルを当てはめ、また帰無仮説における各部分を

推定および検定を行っています。数値の厳密さを保持し、また推定可能でなくなることを

回避するため、DIVISOR=オプションが使用されています。 
 
proc glm data=test; 

  class a b; 

  model y=a b a*b; 

  estimate 'AB12' intercept 1 a 1 0 b 0 1 0 a*b 0 1 0 0 0 0; 

  estimate 'avg ABij' intercept 6 a 3 3 b 2 2 2 

a*b 1 1 1 1 1 1 / divisor=6; 

  estimate 'AB12 vs avg ABij' a 3 -3 b -2 4 -2 

a*b -1 5 -1 -1 -1 -1 / divisor=6; 

run; 

quit; 

 

Parameter Estimate Standard Error t Value Pr > |t| 

AB12 10.0341436 0.25521923 39.32 <.0001 

avg ABij 11.3122544 0.11799152 95.87 <.0001 

AB12 vs avg ABij -1.2781108 0.23947144 -5.34 <.0001 

 
 
 

例題 2：交互作用項を含む 3 要因の分散分析 

 
ここでは、それぞれ水準の数が 5、2、3 である 3 要因のモデルを仮定します。主効果、お



よび全ての交互作用項を含めたモデルは以下のように記述することができます。 
 

Yijkl = μ + αi + βj + γk + αβij + αγik + βγjk + αβγijk + εijkl    (2) 
 
ここで、i=1,2,...,5,  j=1,2,  k=1,2,3,  and l=1,2,...,Nijk とします。 
 
次のようなプログラムによって、このモデルから擬似データセットを作成することができ

ます。 
 
data test; 

seed=8422636; 

  do a=1 to 5; 

    do b=1 to 2; 

      do c=1 to 3; 

        do rep=1 to ceil(ranuni(seed)*3)+3; 

          y=5 + a + b + c + a*b + a*c + b*c + a*b*c + rannor(seed); 

          output; 

        end; 

      end; 

    end; 

  end; 

run; 

 
次のプログラムではモデル式(2)への当てはめを行っており、また solution ベクトルと各セ

ル平均が算出されます。 
 
proc mixed data=test; 

  class a b c; 

  model y=a|b|c / solution; 

  lsmeans a*b*c; 

run; 

 
ABC121 と ABC212 のセル平均が等しいとする帰無仮説を検定するとき、モデル式(2)から

各平均値とその差は以下のように記述できます。 
 

μ121 = μ + α1 + β2 + γ1 + αβ12 + αγ11 + βγ21 + αβγ121  



μ212 = μ + α2 + β1 + γ2 + αβ21 + αγ22 + βγ12 + αβγ212  
μ121 – μ212 = α1 – α2 – β1 + β2 + γ1 – γ2 + αβ12 – αβ21 + αγ11 – 
αγ22 + βγ21 – βγ12 + αβγ121 – αβγ212  

 
次のプログラムは、それぞれの平均とその差の推定値を求め、差が 0 であるかを検定して

います。LSMEANS ステートメントを用いても同様の推定、および検定を行うことができ

ます。 
 
proc mixed data=test; 

  class a b c; 

  model y=a|b|c / solution; 

  lsmeans a*b*c / pdiff; 

  estimate 'ABC121' intercept 1 a 1 b 0 1 c 1 

                        a*b 0 1 a*c 1 b*c 0 0 0 1 

                        a*b*c 0 0 0 1; 

  estimate 'ABC212' intercept 1 a 0 1 b 1 c 0 1 

                        a*b 0 0 1 a*c 0 0 0 0 1 b*c 0 1 

                        a*b*c 0 0 0 0 0 0 0 1; 

  contrast 'ABC121 - ABC212' a 1 -1 b -1 1 c 1 -1 

                               a*b 0 1 -1 a*c 1 0 0 0 -1 

                                  b*c 0 -1 0 1 

                                  a*b*c 0 0 0 1 0 0 0 -1; 

  estimate 'ABC121 - ABC212' a 1 -1 b -1 1 c 1 -1 

                                  a*b 0 1 -1 a*c 1 0 0 0 -1 

                                  b*c 0 -1 0 1 

                                  a*b*c 0 0 0 1 0 0 0 -1; 

run; 

 

Estimates 

Label Estimate Standard Error DF t Value Pr > |t| 

ABC121 17.0454 0.4118 125 41.40 <.0001 

ABC212 21.8270 0.4118 125 53.01 <.0001 



ABC121 - ABC212 -4.7816 0.5823 125 -8.21 <.0001 

 

Contrasts 

Label Num DF Den DF F Value Pr > F 

ABC121 - ABC212 1 125 67.42 <.0001 

 

Least Squares Means 

Effect a b c Estimate Standard Error DF t Value Pr > |t| 

a*b*c 1 2 1 17.0454 0.4118 125 41.40 <.0001 

a*b*c 2 1 2 21.8270 0.4118 125 53.01 <.0001 

 

Differences of Least Squares Means 

Effect a b c _a _b _c Estimate Standard Error DF t Value Pr > |t|

a*b*c 1 2 1 2 1 2 -4.7816 0.5823 125 -8.21 <.0001

 
 

例題 3：交互作用項を含む 2 要因のロジスティック回帰モデル 

 
ロジスティックモデルは、一般化線形モデルの一つとなります。これらのモデルに対して

は、応答変数が直接モデル化されることはありません。その代わりに、応答分布の平均の

関数としてモデルを記述することになります。ロジスティックモデルでは、応答変数の分

布は 2 項分布であり、対数オッズ(または、2 項分布の平均 p のロジット(logit))を応答関数

としてモデルを記述することになります。 
 

logit(pi) ≡ log(Oddsi) ≡ log[pi / (1–pi)] 
 
ロジスティック回帰モデルに関する更なる情報については、本文献の最後をご参照くださ



い。 
 
ここでは、「complicated（複雑性）」、または「uncomplicated（単純性）」と診断された患

者に対し、3 つの治療方法 (A,B,およびC)から 1 つを行い、その結果((cured（治癒）、uncured
（未治癒）)を観察した、以下の医学データを例として使用します。2 
 
data uti; 

   input diagnosis : $13. treatment $ response $ count @@; 

   datalines; 

complicated    A  cured 78  complicated   A not 28 

complicated    B  cured 101 complicated   B not 11 

complicated    C  cured 68  complicated   C not 46 

uncomplicated  A  cured 40  uncomplicated A not 5 

uncomplicated  B  cured 54  uncomplicated B not 5 

uncomplicated  C  cured 34  uncomplicated C not 6 

; 

 
 
Dummyコーディング 
 
説明変数に対する indicator または dummy コーディングは、デザイン行列における実際

の変数を、元の変数の水準を示す 0 または 1 の値を持ついくつかの変数の組で置き換えま

す。元の変数の各水準に対して、1 つの変数が作成されることになります。主効果に対する

パラメータは、参照水準(デフォルトでは最後の水準)とその水準の効果の差としてとらえる

ことができます。このコーディングは、GLM プロシジャと MIXED プロシジャの CLASS
変数に対するデフォルトのコーディング方法となります。LOGISTIC プロシジャは、

CLASS ステートメントで PARAM=GLM と追記することによって、CLASS 変数に対して

dummy コーディングとすることができます。フルランクの indicator コーディング

(reference コーディングとも呼ばれます)では、最後の水準に対する indicator 変数を削除す

ることになります。この手法は、LOGISTIC プロシジャや CATMOD プロシジャで利用で

き、他のプロシジャでも PARAM=REF オプションを指定することによって利用可能となり

ます。 
 

                                                  
2 このデータを使用した解析例は、”Categorical Data Analysis Using SAS System”, 
Chapter 8, 8.4 Qualitative Explanatory Variables に記載されています。そこでは、尿路

感染症(urinary tract infection)に関するデータであるとされています。 



応答変数に正規性を仮定した例題 1と同様に、dummyコーディングを使用して、ロジステ

ィックモデルにおける右辺を以下のように記述することができます。 
 

log(Oddsij) = μ + αi + βj + αβij      (3a)  
 
ここで、 i=1,2,...,5,  j=1,2,  k=1, 2,...,Nij  とします。しかし、これは以下のように記述す

ることもできます。 
 

log(Oddsij) = μ + ΣiαiAi + ΣjβjBj + ΣijγijAiBj           (3b)  
 
ここで、Aiと Bjは dummy コーディングを使用した以下のように定義されているデザイン

変数となります。 
 

• A = i のときには Ai = 1、他の場合には Ai = 0 
• B = j のときには Bj = 1、他の場合には Bj = 0 

前記の医学データに対して、モデル(3b)の場合の”cured”（治癒）に対するオッズは以下の

ように記述することになります。 
 

log(Oddsdt) = μ + d1O + d2U + t1A + t2B + t3C + g1OA + g2OB + g3OC + g4UA + 
g5UB + g6UC           (3c)  

 
ここで、 

• O は”complicated”の診断に対するダミー変数  
• U は” uncomplicated”の診断に対するダミー変数  
• A、B、および C は、3 つの治療方法に対するダミー変数 
• OA から UC までは、診断と治療方法の交互作用に対するダミー変数 

です。 
 
 

Dummyコーディングを用いてオッズ比の推定と検定を行う 

 
平均ではなく対数オッズをモデル化していることから、MIXED プロシジャや GLM プロシ

ジャのように平均に関してではなく、対数オッズに関する対比の推定や検定を行うことに



なります。しかし、CONTRAST ステートメントの指定に関しては同じプロセスとなります。

つまり、ステートメントにおける係数を決めるために、当てはめモデルに基づいて、関心

を持っている仮説を書き下します。 
LOGISTIC プロシジャでは、CONTRAST ステートメントにて対比が 0 であるかの検定を

行うともに、対比の推定値も出力されます。そのため、このプロシジャでは ESTIMATE ス

テートメントが用意されていません。 
 
対数オッズ比は、対数オッズの差として表すことができます。 
 

log(Oddsi) – log(Oddsj) = log( Oddsi / Oddsj ) = log(ORij)  
 
つまり、対数オッズの差に対する推定値の指数をとることによって、オッズ比の推定値が

得られます。CONTRAST ステートメントで ESTIMATE=BOTH オプションを指定すると、

対比(対数オッズの差、言い換えると対数オッズ比)と、対比の指数をとった値(オッズ比)の
両方の推定値を表示させることができます。 
 
この医学データの例に対して、”complicated”と診断されている患者における治療方法 A と

C に関するオッズ比に関心があるものとします。このとき、帰無仮説は以下のように書くこ

とができます。 
 

H0: log(OddsOA) = log(OddsOC) 
または 

H0: log(OddsOA) - log(OddsOC) = 0 
 
次のテーブルは、” complicated”と診断されている場合におけるデータを要約したものです。 
 

Table of treatment by response 

response treatment

cured not 

Total

A 78 28 106

C 68 46 114

Total  146 74 220



 
 
データから、オッズ比は以下のようにして計算することができます。 
 
   78/28   78·46 

   ----- = ----- = 1.8845 

   68/46   68·28 

 
つまり、”complicated”と診断された場合、治療 A によって治癒するオッズは、治療 C によ

って治癒するオッズの 1.8845 倍として解釈することができます。上記のテーブルとオッズ

比は、次のようなプログラムを記述することによって表示することができます。 
 
proc freq data=uti; 

  where diagnosis = "complicated" and treatment in ("A","C"); 

  table treatment * response / relrisk norow nocol nopercent; 

  weight count; 

run; 

 
 

Estimates of the Relative Risk (Row1/Row2) 

Type of Study Value 95% Confidence Limits 

Case-Control (Odds Ratio) 1.8845 1.0643 3.3367 

Cohort (Col1 Risk) 1.2336 1.0210 1.4906 

Cohort (Col2 Risk) 0.6546 0.4440 0.9652 

 
モデル(3c)を用いて、対数オッズに関する同様の対比を推定し、検定を行うためには、

CONTRASTステートメントで必要な対比係数を得るために例題 1と同じプロセスを経るこ

とになります。デザイン変数の順序を確認するためには、 ”Analysis of Maximum 
Likelihood Estimates”のテーブルを見てみましょう。この確認は、CONTRASTステートメ

ントにおける係数の適切な順序を考慮する上でとても重要です。この例については、当該

テーブルからモデル(3c)におけるパラメータの順序の通りに並んでいることが確認するこ

とができます。 
 



“complicated”と診断された患者に対する治療 A の対数オッズは、以下のように記述するこ

とができます。 
 

log(OddsOA) = μ + d1 + t1 + g1  
 
また、同様に、治療方法 C の対数オッズは、以下の式となります。 
 

log(OddsOC) = μ + d1 + t3 + g3  
 
上記の差より、対数オッズの差、つまり、対数オッズ比を求めることができます。 
 

log(OddsOA) - log(OddsOC) = t1 - t3 + g1 - g3 
 
次のステートメントではモデル(3c)を当てはめ、対比を推定しています。対比の推定値を指

数化することによって、オッズ比の推定値を得ることができます。 
 
proc logistic data=uti; 

  freq count; 

  class diagnosis treatment / param=glm; 

  model response = diagnosis treatment diagnosis*treatment; 

  contrast 'trt A vs C in comp' treatment 1 0 -1 

            diagnosis*treatment 1 0 -1 0 0 0 / estimate=both; 

  output out=out xbeta=xbeta; 

run; 

 

Contrast Rows Estimation and Testing Results 

Contrast Type Row Estimate 

Standard

Error Alpha

Lower 

Limit

Upper 

Limit

Wald 

Chi-Square Pr > ChiSq

trt A vs 

C in 

comp 

PARM 1 0.6336 0.2915 0.05 0.0623 1.2050 4.7246 0.0297

trt A vs 

C in 
EXP 1 1.8845 0.5493 0.05 1.0643 3.3367 4.7246 0.0297



comp 

 
 
OUTPUT ステートメントにおける XBETA=オプションは、各オブザベーションにおける

線形予測子 x’β を求めており、対数オッズ比に相当します。以下のプログラム

は、”complicated”との診断における、治療方法 A と治療方法 C の対数オッズをそれぞれ推

定していることになります。 
 
proc print data=out noobs; 

  where diagnosis="complicated" and response="cured" and 

         treatment in ("A","C"); 

  var diagnosis treatment xbeta; 

run; 

 
 

diagnosis treatment xbeta 

complicated A 1.02450

complicated C 0.39087

 
2 つのセルに対数オッズの差、(1.02450 – 0.39087 = 0.63363)は、CONTRAST ステートメ

ントにて求めた対数オッズ比と同じになっています。また、この値の指数をとると 
(exp[0.63363] = 1.8845)、CONTRAST ステートメントから得られた、対比の指数をとった

数値(オッズ比の推定値)となっています。 
 

枝分かれ（ネステッド）モデル(Nested Model) 

 
主効果と交互作用項を含むモデル(3c)ではなくても、同じことは以下のように枝分かれモデ

ルを用いて実現できます。 
 

log(Oddsdt) = μ + dd + t(d)dt             
または、 

log(Oddsdt) = μ+d1O+d2U+g1OA+g2OB+ g3OC + g4UA +g5UB+g6UC           (3d)  
 



枝分かれ項(nested term)では、前のモデルにおける「治療」と交互作用項の自由度と同じ

値となります。枝分かれ項を生成するデザイン変数は、すでに作成した交互作用項に対す

るものと同じになります。しかしながら、枝分かれモデルでは、診断の各水準内における

治療の対比がより明らかとなります。この点に関するさらなる詳細、どのようにデザイン

行列が作成されるかに関する詳細、については、SAS/STAT User’s Guide，GLM プロシジ

ャにおける”Parameterization of PROC GLM Models”の節をご参照ください。ここまで記

述してきたように、各効果に対するデザイン変数の順序を知ることは、CONTRAST ステー

トメントにて対比係数を適切に指定する上でとても重要となります。 
 
枝分かれモデル(3d)を用いると、対数オッズの対比は以下のように記述できます。 
 

log(OddsOA) = μ + d1 + g1  
 
また、「複雑性」の診断における治療方法 C に対する対数オッズは、以下のようになります。 
 

log(OddsOC) = μ + d1 + g3  
 
これらの差を求めることにより、対数オッズの差、つまり対数オッズ比が得られます。 
 

log(OddsOA) – log(OddsOC) = g1 – g3 
 
この対比は、枝分かれ効果における主効果(治療)の部分の対比、つまり、治療の水準とその

最後(参照)水準の比較に相当します。そのため、この対比は、ネステッド効果における診断

(diagnosis)の水準内における治療 A のパラメータとしても推定することができます。次の

ステートメントでは、ネステッドモデルの推定を行い、対比を算出していることになりま

す。”Analysis of Maximum Likelihood Estimates”の表にて、モデル(3d)におけるデザイン

変数の順序を確認することができます。枝分かれ効果に対する最初の 3 つのパラメータは、

複雑性(complicated)の診断(diagnosis)の場合における各治療(Treatment)に相当します。次

の 3 つのパラメータは、単純性(uncomplicated)の診断(diagnosis)の場合に相当します。

EXPB オプションを追記することによって、パラメータ推定値の表においてパラメータ推

定値を指数化した値を表示させることができます。 
 
proc logistic data=uti; 

  freq count; 

  class diagnosis treatment / param=glm; 

  model response = diagnosis treatment(diagnosis) / expb; 



  contrast 'trt A vs C in comp' 

             treatment(diagnosis) 1 0 -1 0 0 0 / estimate=both; 

run; 

 
対数オッズ比とオッズ比の推定値の表は、前と同じになります。この場合、複雑性

(complicated)の診断(diagnosis)の場合における治療Aに対するパラメータ推定値は、対比に

おける推定値と同じになり、パラメータ推定値を指数化した値は対比の推定値を指数化し

た値と同じとなっています。 
 
 

Analysis of Maximum Likelihood Estimates 

Parameter     DF Estimate

Standard

Error

Wald 

Chi-Square Pr > ChiSq Exp(Est)

Intercept     1 1.7346 0.4428 15.3451 <.0001 5.667

diagnosis complicated   1 -1.3437 0.4822 7.7653 0.0053 0.261

diagnosis uncomplicated   0 0 . . . .

treatment(diagnosis) A complicated 1 0.6336 0.2915 4.7246 0.0297 1.884

treatment(diagnosis) B complicated 1 1.8262 0.3705 24.3005 <.0001 6.210

treatment(diagnosis) C complicated 0 0 . . . .

treatment(diagnosis) A uncomplicated 1 0.3448 0.6489 0.2824 0.5952 1.412

treatment(diagnosis) B uncomplicated 1 0.6445 0.6438 1.0020 0.3168 1.905

treatment(diagnosis) C uncomplicated 0 0 . . . .

 
 
 
Effectsコーディング 
 
予測変数に対する Effects、または Deviation from mean(平均からの差)コーディングは、



デザイン行列(またはモデル行列)における変数を、元の変数の水準を示す(-1，0，1)の値を

用いた変数で置き換えます。作成される変数の数は、元の変数における水準の数より１少

ない数となります。その結果、推定されているパラメータの数も水準の数より一つ少ない

ことになります。主効果に対するパラメータは全ての水準の平均的な効果からの差異とし

て捉えることができます。このコーディングは、CATMOD プロシジャ、LOGISTIC プロ

シジャにおけるデフォルトの手法であり、他のプロシジャにおいても PARAM=EFFECT
オプションを指定することによって用いることができます。 
 
Effectsコーディングの場合には、モデルは依然として(3b)のように記述されます。しかしな

がら、診断(diagnosis)、治療(treatment)に対するデザイン行列は、次の表のように異なっ

た定義となります。diagnosis(もしくはtreatments)に対する行を合計した場合、0 となりま

す。 
Effects コーディングを使用した場合、パラメータは合計が 0 という制約が与えられている

ため、効果の最後の水準に対する推定値は、αa= –(α1 + α2 + ... + αa–1)として表される

ことになります。 
 
 

Class Level Information 

Design VariablesClass Value 

1 2

diagnosis complicated 1  

  uncomplicated -1  

treatment A 1 0

  B 0 1

  C -1 -1

 
 

Effectsコーディングによるダミー変数を用いて、オッズ比の推定と検定を行う 

 
コーディング方法は異なりますが、同じようなステップを踏むことによって対比の係数を



求めることができます。最初にモデルを記述し、プロシジャが表示する結果よりパラメー

タとその順序を確認します。 
 

log(Oddsdt) = μ + dO + t1A + t2B + g1OA + g2OB           (3e)  
 
Effectsコーディングによるモデル(3e)に基づき、対比における各部分を記述します。この場

合、”Class Level Information”テーブルがデザイン行列の設定を確認する上で有用で

す。”complicated”の診断、治療Aの場合には、O=1、A=1、B=0 となります。このため、対

数オッズは以下のようになります。 
 

log(OddsOA) = μ + d + t1 + g1  
 
complicated の診断、治療 C の場合には、O=1、A=-1、B=-1 となります。このため、対数

オッズは以下のように記述されます。 
 

log(OddsOC) = μ + d – t1 – t2 – g1 – g2  
 
これらの差を求めることにより、対数オッズの差、つまりオッズ比を次のように記述する

ことができます。 
 

log(OddsOA) – log(OddsOC) = 2t1 + t2 + 2g1 + g2 
 
次のステートメントでは、Effects 法のコーディングを用いたモデルを推定し、対比を推定

しています。 
 
proc logistic data=uti; 

  freq count; 

  class diagnosis treatment; 

  model response = diagnosis treatment diagnosis*treatment; 

  contrast 'trt A vs C in comp' treatment 2 1 

            diagnosis*treatment 2 1 / estimate=both; 

run; 

 
Glm 法によるダミー変数を用いたモデルと、同じ対数オッズ比、およびオッズ比の推定値

が得られています。 
 



Effectコーディングに基づいて、より複雑なCONTRASTステートメントの指定を

行う 

 
次に、複雑性(complicated)，治療 A における効果と、治療における平均的な効果が同じで

あるかを検証するとします。この場合、モデル(3e)を用いると帰無仮説は以下のようになり

ます。 
 

H0: log(OddsOA) = Σijlog(Oddsij)/3 
 

もしくは 
 

H0: log(OddsOA) – Σijlog(Oddsij)/3 = 0 
 
すでに記述しましたように、帰無仮説における最初の部分は、log(OddsOA) = μ + d + t1 + 
g1、となります。平均的な治療の効果は、次のように記述することになります。 
 

Σijlog(Oddsij)/3 = [(μ + d + t1 + g1) + (μ + d + t2 + g2) + (μ + d – t1 – t2 – g1 – 
g2)]/3 = (3μ + 3d)/3 = μ + d 

 
この式を log(OddsOA):から引くことによって、以下の式を求めることができます。 
 

(μ + d + t1 + g1) – (μ + d) = t1 + g1 
 
このことから、次のように CONTRAST ステートメントを記述すると、対比の推定、およ

び帰無仮説の検定を行うことができます。結果として、以下のような表がアウトプットウ

ィンドウに表示されます。 
 
  contrast 'trt A vs avg trt in comp' treatment 1 0 

            diagnosis*treatment 1 0 / estimate=both; 

 
 
 

Contrast Rows Estimation and Testing Results 

Contrast Type Row Estimate 

Standard

Error Alpha

Lower 

Limit

Upper 

Limit

Wald 

Chi-Square Pr > ChiSq



Contrast Rows Estimation and Testing Results 

Contrast Type Row Estimate 

Standard

Error Alpha

Lower 

Limit

Upper 

Limit

Wald 

Chi-Square Pr > ChiSq

trt 1 vs 

avg trt 

in comp 

PARM 1 -0.1863 0.1919 0.05 -0.5624 0.1898 0.9428 0.3316

trt 1 vs 

avg trt 

in comp 

EXP 1 0.8300 0.1593 0.05 0.5698 1.2090 0.9428 0.3316

 
 
対比推定値の指数をとった値 0.83 は、実際にはオッズ比ではありません。指数をとった場

合、分母の部分は治療のオッズに対する幾何平均に相当します。正確にはオッズ比ではあ

りませんが、「治療 A」と「治療の平均的な効果」を比較する上で、この数値は有用なもの

です。 
 
CATMOD プロシジャも Effects コーディングを用いているため，この CONTRAST ステー

トメントを用いて、上記と同様の結果を得ることができます。CATMOD プロシジャで

WEIGHT ステートメントを用いることによって、各セルの度数として要約されているデー

タを入力データセットとして用いることができます。 
 
proc catmod data=uti; 

  weight count; 

  model response = diagnosis treatment diagnosis*treatment; 

  contrast 'trt 1 vs avg trt in comp' treatment 1 0 

             diagnosis*treatment 1 0 / estimate=both; 

run; 

 
CATMOD プロシジャには、この種の検定をより簡略に行う機能があります。つまり、

MODEL ステートメントで、「値ごとの枝分かれ効果(nested-by-value effects)」の形で指定

することができます。この指定により、他の変数のある水準内における変数の効果を検証

することができます。これは、GLM プロシジャや LOGISTIC プロシジャなど、他のプロ

シジャで指定できる枝分かれモデルの延長となります。この点に関する更なる情報は、



SAS/STAT User’s Guide で CATMOD プロシジャの章の”Generation of the Design 
Matrix”の項をご参照ください。 
 
医学データのサンプルの場合、treatment*diagnosis という交互作用項を「値ごとの枝分

かれ効果」を使用して以下のように分割することができます。 
 
proc catmod data=uti; 

  weight count; 

  model response = diagnosis treatment(diagnosis='complicated') 

                      treatment(diagnosis='uncomplicated'); 

run; 

 
treatment(diagnosis='complicated') と treatment(diagnosis='uncomplicated') の２つの

効果は、診断(diagonosis)の各水準内において、治療の効果を検証している nested-by-value 
effects の指定に相当します。以下の出力では、treatment(diagnosis='complicated')の効果

に対する最初のパラメータは、”complicated”の診断における、治療方法 A の効果と治療の

平均的な効果の比較を検証していることになります。これは、前で作成した対比とまさし

く同じです。 
 

Analysis of Maximum Likelihood Estimates 

Parameter   Estimate Standard

Error

Chi- 

Square 

Pr > ChiSq 

Intercept   1.6377 0.1514 116.98 <.0001 

Diagnosis complicated -0.4268 0.1514 7.95 0.0048 

treat(diagn=complicated) A -0.1864 0.1919 0.94 0.3315 

  B 1.0064 0.2329 18.67 <.0001 

trea(diag=uncomplicated) A 0.0149 0.3822 0.00 0.9689 

  B 0.3150 0.3793 0.69 0.4063 

 
 



例題 4. モデルの比較 – 尤度比検定 

 
Ries and Smith (1963)の例では、洗剤のブランド(Brand= M または X)の選択が、3 つの

他のカテゴリ変数に関連していることが記載されています。他の変数とは、洗濯に用いる

水の硬度(Softness=soft（軟）, medium（中間）, または hard（硬）)、水温(Temperature= 
high（高） または low（低）)、ブランド M を以前に使ったことがあるか(Previous= yes
（はい）または no（いいえ）) の 3 つです。この例では、2 つのロジスティックモデルを

使用します。1 つ目は、飽和モデル、言い換えると考えられる全ての主効果と交互作用を含

めたモデルであり、このとき全ての利用可能な自由度がモデルの自由度となります。2 つ目

のモデルは、主効果のみからなる「縮小した」モデルです。次のプログラムはデータセッ

トを作成し、飽和ロジスティックモデルを当てはめています。 
 
data detergent; 

  input Softness $ Brand $ Previous $ Temperature $ Count @@; 

  datalines; 

soft X yes high 19   soft X yes low 57 

soft X no  high 29   soft X no  low 63 

soft M yes high 29   soft M yes low 49 

soft M no  high 27   soft M no  low 53 

med  X yes high 23   med  X yes low 47 

med  X no  high 33   med  X no  low 66 

med  M yes high 47   med  M yes low 55 

med  M no  high 23   med  M no  low 50 

hard X yes high 24   hard X yes low 37 

hard X no  high 42   hard X no  low 68 

hard M yes high 43   hard M yes low 52 

hard M no  high 30   hard M no  low 42 

; 

 

ods select modelfit type3; 

ods output modelfit=full; 

proc genmod data=detergent; 

  class Softness Previous Temperature; 

  freq Count; 

  model Brand = Softness|Previous|Temperature / dist=binomial type3; 



run; 

 
結果の一部からすると、このモデルでは交互作用項が必要ないことが示唆されています。 
 

Criteria For Assessing Goodness Of Fit 

Criterion DF Value Value/DF 

Deviance 996 1364.4956 1.3700 

Scaled Deviance 996 1364.4956 1.3700 

Pearson Chi-Square 996 1008.0000 1.0120 

Scaled Pearson X2 996 1008.0000 1.0120 

Log Likelihood   -682.2478   

 

LR Statistics For Type 3 Analysis 

Source DF Chi-Square Pr > ChiSq 

Softness 2 0.10 0.9522 

Previous 1 22.13 <.0001 

Softness*Previous 2 3.79 0.1506 

Temperature 1 3.64 0.0564 

Softness*Temperature 2 0.20 0.9066 

Previous*Temperature 1 2.26 0.1327 

Softne*Previo*Temper 2 0.74 0.6917 

 
 
主効果のみからなるモデルは、より簡単に次のように当てはめることができます。 
 



ods select modelfit type3; 

ods output modelfit=reduced; 

proc genmod data=detergent; 

  class Softness Previous Temperature; 

  freq Count; 

  model Brand = Softness Previous Temperature / dist=binomial type3; 

run; 

 

Criteria For Assessing Goodness Of Fit 

Criterion DF Value Value/DF 

Deviance 1003 1372.7236 1.3866 

Scaled Deviance 1003 1372.7236 1.3866 

Pearson Chi-Square 1003 1007.9360 1.0049 

Scaled Pearson X2 1003 1007.9360 1.0049 

Log Likelihood   -686.3618   

 
 

LR Statistics For Type 3 Analysis 

Source DF Chi-Square Pr > ChiSq

Softness 2 0.22 0.8976 

Previous 1 19.89 <.0001 

Temperature 1 3.74 0.0532 

 
ここでの問題は、これらの 2 つのモデル間に統計的に有意な差があるかないかということ

です。別の表現では、主効果のみを含むモデル、つまり交互作用に対するパラメータが 1
つでも 0 とは異なるかを検証することと同じになります。この検定を行うには、2 つのやり

方があります。1 つ目は、各モデルから計算される対数尤度の値から、尤度比検定を構成す



る方法です。もう 1 つは、CONTRAST ステートメントを使用して、交互作用に対するパ

ラメータに関して「同時に」検定する方法です。 
 
 
尤度比検定を構成する 
 
主効果のみのモデルは、いくつかのパラメータ(交互作用項に対するパラメータ)が 0 となっ

ている飽和モデルであることから、上記の 2 つのモデルは「ネストしている」と考えるこ

とができることに着目してください。モデル尤度の比を 2 倍、または同義かつ便利な表現

をすると、対数尤度の差の 2 倍である尤度比(LR)検定統計量を計算して、ネストしている 2
つのモデルを比較することができます。これらのモデルに対しては、対数尤度の差の 2 倍

は 8.228 となります。LR統計量は、検定しているパラメータの数と等しい自由度を持つカ

イ 2 乗分布に従います。この例では、4 つの交互作用項からすると、自由度は 7 となります。

LR統計量に対するp値を得るためには、DATAステップにおいてPROBCHI関数3を使用し

てください。次のプログラムは、主効果のみのモデルと飽和モデルを比較するLR検定に対

するp値を算出するものです。 
 
data lrt; 

   lr=2*(-682.2478 - -686.3618); 

   df=7; 

   p=1-probchi(lr,df); 

run; 

proc print noobs; 

   format p pvalue.; 

run; 

 
この検定における帰無仮説は、飽和モデルと主効果のみのモデルが同等であるということ

です。この例において帰無仮説が真であるためには、交互作用項に対応する全てのパラメ

ータが 0 でなくてはなりません。このため、帰無仮説は「モデルは同等である」となりま

すが、この例では交互作用項がない、と捉えることもできます。結果からすると、帰無仮

説は棄却することができないという解釈になります。(p=.3129) 
 
 

                                                  
3 CDF 関数や、SDF 関数を使用して計算することも可能です。1-probchi(lr.df)は、

1-cdf(‘chisquared,lr,df ’)、sdf(‘chisquared,lr,df ’)と同等です。 



lr df p 

8.228 7 0.3129

 
LR 検定は、最尤法により当てはめた 2 つのネストしたモデルを比較するために使用できま

す。大きな方のモデルが飽和している必要はありません。従って、GENMOD, LOGISITIC, 
MIXED, PHREG, PROBIT など、数多くのプロシジャでこの検定を使うことができます。 
ただし、LR 検定ではネストしていないモデルを比較することはできないことに留意する必

要があります。また、GENMOD プロシジャで REPEATED ステートメントを使用して当

てはめたモデルは、一般化推定方程式(GEE)による方法であり、最尤法ではないことにも注

意してください。同様に、GLIMMIX プロシジャの RANDOM ステートメントを使用して

当てはめたモデルも、本来の対数尤度を使用したものではありません。そのため、この種

のモデルを比較するために LR 検定を使用することはできません。 
 
数値を再度記述することなく4尤度比検定を行いたい場合には、2 つのGENMODプロシジ

ャの結果からModelFitテーブルにおける統計量を用いて算出することになります。

ModelFitテーブルからなるFULL、およびREDUCEDというデータセットを作成するため

に、前記のようにGENMODプロシジャを実行するときにはODS OUTPUTステートメント

を指定してください。 
 
data lrt; 

  retain dff dfr LRDF; 

  set full end=endf; 

  dff=df; 

  if endf then llf=value; 

  set reduced end=endr; 

  dfr=df; 

  if endr then llr=value; 

  if _n_=1 then LRDF=dfr-dff; 

  if endf and endr then do; 

    LR=2*(llf-llr); 

    p=1-probchi(LR, LRDF); 

    keep LR LRDF p; 

    output; 

                                                  
4 ここでは、-682.2478 と-686.3618 という 2 つの数値のこと。 



  end; 

run; 

 
モデルを比較するために「対比」を記述する 
 
上記では、2 つのモデル、飽和モデルと縮小したモデル、を推定し、DATA ステップにてモ

デルの比較を行う際の検討統計量の算出が必要としました。しかしながら、このようなス

テップを行わずに同様の検証を行うことができます。この場合、飽和モデルのみを推定し、

CONTRAST ステートメントにて、縮小したモデルとなるように、同時にパラメータを 0
にすることを検定することになります。 
 
モデルパラメータの推定可能である線形結合は、プロシジャのCONTRASTステートメント

を用いて検定するができます。しかし、交互作用全てのパラメータを同時に検定、かつ推

定可能な線形式の組合せ指定に難しさがあります。とりわけ、indicator(dummy)コーディ

ングの場合には難しくなります。この問題は、フルランクのコーディングを用いてかなり

単純化することができます。この場合、CLASSステートメントでPARAM=EFFECT5オプ

ションを指定することにより、このコーディングを用いることができます。CONTRASTス
テートメントでは、Lβ=0 の帰無仮説として検定を行うことができます。ここで、Lは帰無

仮説を表現する行列であり、βはモデルパラメータからなるベクトルです。Effectsコーデ

ィングにより、Lの各行はβのベクトルと掛け合わせるときに、それぞれ一つの交互作用の

パラメータを選択するように記述します。CONTRASTステートメントでは、Lの行をカン

マで区切られます。下記のCONTRASTステートメントでは、行列Lは 7 行から構成され、

各行はそれぞれの交互作用に対応しています。また、合計で自由度は 7 となります。 
 
ods select contrasts; 

proc genmod data=detergent; 

  class Softness Previous Temperature / param=effect; 

  freq Count; 

  model Brand = Softness|Previous|Temperature / dist=binomial; 

  contrast 'lrt' 

  softness*previous 1 0, 

  softness*previous 0 1, 

  softness*temperature 1 0, 

  softness*temperature 0 1, 

                                                  
5 GENMOD プロシジャでは、SAS9 以降で使用できる機能です。 



  previous*temperature 1, 

  softness*previous*temperature 1 0, 

  softness*previous*temperature 0 1; 

run; 

 
結果は、前項で計算された LR 統計量と同じになります。これは、GENMOD プロシジャで

は、指定した対比に対して LR 統計量がデフォルトで算出されるからです。 
 

Contrast Results 

Contrast DF Chi-Square Pr > ChiSq Type

lrt 7 8.23 0.3129 LR 

 
 
LOGISTIC プロシジャなどのいくつかのプロシジャでは、LR 統計量ではなく Wald カイ 2
乗統計量を算出します。GENMOD プロシジャでは、CONTRAST ステートメントで WALD
オプションを指定すると WALD 統計量が算出されます。LR 統計量と Wald 統計量は、漸

近的に等価です。 
 
ods select contrasttest; 

proc logistic data=detergent; 

  class Softness Previous Temperature / param=effect; 

  freq Count; 

  model Brand = Softness|Previous|Temperature; 

  contrast 'lrt' 

  softness*previous 1 0, 

  softness*previous 0 1, 

  softness*temperature 1 0, 

  softness*temperature 0 1, 

  previous*temperature 1, 

  softness*previous*temperature 1 0, 

  softness*previous*temperature 0 1; 

run; 

 
 



Contrast Test Results 

Contrast DF Wald 
Chi-Square

Pr > ChiSq

lrt 7 8.1794 0.3170 

